Cold Spray for Grins

Ruiyu Ou

August 8, 2026

Contents

1 Introduction

2 Gen AI Usage Disclosure

All physics and coding is done by myself. Gen AI platforms such as Claude or ChatGPT are used to expedite reference searching, and deal with debugging. This LaTeX code still does not work properly despite Gen AI working on it. Maybe because I am using the free versions so they can't handle my disgustingly poor LaTeX code (may be more due to how I structured the dirs).

3 Free Boundary Problem

3.1 The Stefan Problem

3.1.1 Description and Connection with Cold Spray

The Stefan problem describes heat transfer in a material containing a moving phase boundary, where the interface between two phases evolves over time as latent heat is absorbed or released. A classic example is the melting or freezing of ice submerged in a body of water, where the solid–liquid interface moves in response to heat conduction.

A particularly important engineering application of Stefan-type problems is welding. In conventional welding processes, metal at the joint interface is heated above its melting temperature, with or without the addition of a filler material, before solidifying to form a strong metallurgical bond. Cold spray deposition shares several characteristics with explosive welding, although the underlying mechanisms differ. During cold spray, metallic particles impact the substrate at supersonic velocities, producing severe plastic deformation and localized adiabatic heating. These conditions promote adiabatic shear instability and oxide-layer disruption, enabling solid-state metallurgical bonding between the particles and the substrate without bulk melting.[**?**]

3.1.2 The Mathematical Problem

We will consider the 2 phase Stefan problem in 2 dimensions. [?]

We denote with Ω_i as the region for each phase i, $T_i(\vec{x}, t)$ as the temperature in the respective region, $\partial\Omega_i$ as the boundaries, α_i as the thermal diffusivity, and Q_i as sources of heat.

Within each region, it will satisfy the inhomogeneous heat diffusion equation with their respective thermal diffusivity and Robin boundary conditions. One specific condition is that the temperature at the free boundary be equivalent to the critical phase transition temperature.

$$\frac{\partial T_i}{\partial t} = \nabla \cdot (\alpha_i \nabla T_i) + Q_i \text{ ; for } \vec{x} \in \Omega_i \tag{1}$$

$$a(x)T_i(x) + b(x)\partial_n T_i(x) = g(x) \text{ ; for } \vec{x} \in \partial\Omega_i \tag{2}$$

The interface transitions between each phases i and j with a free surface velocity $\vec{V}$ that satisfies the following differential equation.

$$\vec{V}_i(x,t) = \frac{\mathrm{d}x}{\mathrm{d}t} \text{ ; for } \vec{x} \in \partial\Omega_i \tag{3}$$

$$L_{ij}\vec{V}_i = \alpha_i \partial_{n_i} T_i - \alpha_j \partial_{n_j} T_j \text{ ; for } \vec{x} \in \partial\Omega_i \cap \partial\Omega_j \tag{4}$$

With L_{ij} being the latent heat of phase transition between phases i and j, and ∂_{n_i} being the limit of the outward normal derivative of Ω_i.

3.2 FreeFEM Implementation

3.2.1 The Chosen Toy Problem

We choose the classic problem of ice melting and solidifying. CGS units are used here.

The necessary constants are stated in the table below.

Name	Notation in code	Numerical value
Thermal diffusivity of ice	k_{ice}	1.335E-2 cm^2/s
Thermal diffusivity of water	k_{water}	1.4E-3 cm^2/s
Latent heat of melting	L	334E-2 cm^2/s^2

Where we have chosen to scale the latent heat of melting by 1E-11 to amplify the effects of melting for visual purposes.

We consider the problem in 1cm by 1cm square bisected by the given parametric function.

$$x(t) = 0.3 + 0.4\frac{\frac{1}{1+\exp(-k(t-0.5))} - \frac{1}{1+\exp(0.5k)}}{\frac{1}{1+\exp(-0.5k)} - \frac{1}{1+\exp(0.5k)}} \tag{5}$$

$$y(t) = 1 - t \tag{6}$$

where the parameter k determines the smoothness of the jump. The region on the left will be ice, and the region on the right will be water. See Fig **??**



Figure 1: The curve bisecting the 2 phases. The region on the left will be ice, and the region on the right will be water.

We choose the below boundary conditions.

$$\frac{\partial T_{ice}}{\partial n} = 0 \text{ ; for (x,y) on the left, upper, and lower edges.} \tag{7}$$

$$T_{ice} = 0 \text{ ; for (x,y) on the free surface.} \tag{8}$$

$$\frac{\partial T_{water}}{\partial n} = 0 \text{ ; for (x,y) on the upper and lower edges.} \tag{9}$$

$$T_{water} = 0 \text{ ; for (x,y) on the free surface.} \tag{10}$$

$$T_{water} = 5 \text{ ; for (x,y) on the right edge.} \tag{11}$$

The initial temperatures will be -1 for ice, and 1 for water everywhere. Finally, with no external heat source $- Q = 0$.

3.2.2 The Numerical Implementation

We build the meshes separately to obtain the normal vectors at the boundaries, and assign the appropriate regions as ice and water by use of an indicator function "regIce".

```
// Left region (ice)
mesh ThL = buildmesh(S1(n) + S2(n) + S3(n) + S4(n) + D1(n));

// Isolate left region (ice): region containing point (0.1,
    0.5)
int regIce = ThL(0.1, 0.5).region;
mesh ThIce   = trunc(ThL, region == regIce);
mesh ThWater = trunc(ThL, region != regIce);
```

```
mesh IceStart= ThIce;
```

We then build the piecewise quadratic finite element space.

```
fespace ThI(ThIce, P2);
ThI dxI,dyI, i1, i2; // Mesh moving fespace
ThI iu, iuold, iv; // temperature fespace

fespace ThW(ThWater, P2);
ThW dxW,dyW, w1, w2;
ThW wu, wuold, wv;
```

Where we keep the old temperature to use the Crank-Nicholson solver.

We then solve the homogeneous heat diffusion problem with Crank-Nicholson.

```
    real theta = 0.5; // Crank-Nicholson

problem HeatICN(iu, iv)
    = int2d(ThIce)(
          iu*iv/dt
        + theta * kice * (
              dx(iu)*dx(iv)
            + dy(iu)*dy(iv)
        )
    )
    - int2d(ThIce)(
          iuold*iv/dt
        - (1-theta) * kice *(
              dx(iuold)*dx(iv)
            + dy(iuold)*dy(iv)
        )
    )
    + on(diag,  iu=0)
    ;
problem HeatWCN(wu, wv)
    = int2d(ThWater)(
          wu*wv/dt
        + theta * kwater * (
              dx(wu)*dx(wv)
            + dy(wu)*dy(wv)
        )
    )
    - int2d(ThWater)(
          wuold*wv/dt
        - (1-theta) * kwater *(
              dx(wuold)*dx(wv)
            + dy(wuold)*dy(wv)
        )
    )
    + on(right, wu=5)
    + on(diag,  wu=0)
```

```
    ;
```

We also set up a Laplace problem for moving the boundary, which will be elaborated upon in the next subsubsection.

```
    problem MoveIX(dxI, i1) =
    int2d(ThIce)(dx(dxI)*dx(i1) + dy(dxI)*dy(i1))
  + on(left, dxI = 0)
  + on(diag,  dxI = dt/L*(kice*dx(iu)*N.x-kwater*dx(wu)*N.x)
      )
  ;

problem MoveIY(dyI, i2) =
    int2d(ThIce)(dx(dyI)*dx(i2) + dy(dyI)*dy(i2))
  + on(diag,  dyI = dt/L*(kice*dy(iu)*N.y-kwater*dy(wu)*N.y)
      )
  + on(top, dyI = 0)
  + on(bottom, dyI = 0)
  ;

problem MoveWX(dxW, w1) =
    int2d(ThWater)(dx(dxW)*dx(w1) + dy(dxW)*dy(w1))
  + on(right, dxW = 0)
  + on(diag,  dxW = -dxI)
  ;

problem MoveWY(dyW, w2) =
    int2d(ThWater)(dx(dyW)*dx(w2) + dy(dyW)*dy(w2))
  + on(diag,  dyW = -dyI)
  + on(top, dyW = 0)
  + on(bottom, dyW = 0)
  ;
```

3.2.3 Challenge with Moving the Boundary

We quickly ran into a challenge with moving the boundary.

The numerical formulation for the boundary move increment Δx is given by the following equation.

$$\Delta x_i = \Delta t \cdot \vec{V} = \frac{\Delta t}{L}\left(\alpha_i \partial_{n_i} T_i - \alpha_j \partial_{n_j} T_j\right) \tag{12}$$

The first numerical formulation on incrementing the boundary only moved the finite element points on the boundary. This causes some of the finite elements to flip and FreeFEM terminates with an error of negative area.

To resolve the error, one has a few choices. The simplest way forward is to adjust the time step such that the boundary does not move enough to flip the finite elements, and then remesh to eliminate bad mesh geometry at the boundary. However, this approach severely restricts the time discretization stepsize by the size of the finite elements, and is difficult to predict the appropriate step size.

Naturally then, the choice will be to move all points, and remesh whenever necessary. The problem now is to interpolate the increment everywhere within the domain given the boundary increment. Thus, we choose to solve the following Laplace problem.

$$\nabla^2 \vec{X}_i = 0 \quad \text{in } \Omega_i \tag{13}$$

$$\vec{X}_i = \Delta x_i \quad \text{on } \partial\Omega_i \cap \partial\Omega_j \tag{14}$$

Where $\vec{X}_i$ will be the mesh increment everywhere in Ω_i. Given all the regularization theorems regarding the Laplace operator, one can guess that this allows a much wider range of time step sizes to be used. We present no analysis on the validity of this guess here.

3.2.4 Results

The results can be visualized as a time series in Paraview, or any .vtu viewer.

The ice region grows in size as it solidifies due to a higher thermal diffusivity, as it loses heat energy into the water region. It then stops growing and begins to melt as the water which is maintained at 5 degree celsius on the right bound inputs heat into the ice region.

In the end, we get a net loss of ice region.

4 Contact Mechanics

4.1 The Contact Problem

4.1.1 Description and Connection with Cold Spray

Contact problems describe the mechanical interaction between two or more deformable bodies that come into contact. The primary challenge is determining the contact interface while enforcing the non-penetration constraint. Depending on the physical model, frictional effects may also be incorporated to describe tangential resistance and sliding along the contact surface.

In cold spray additive deposition, metallic particles impact a target substrate at supersonic velocities, producing severe plastic deformation and localized adiabatic heating. These conditions promote adiabatic shear instability amongst other effects, which allows the particle to then metallurgically bond to the substrate material. [?] Particle-to-subtrate and particle-to-particle bonding are both present in cold spray deposition, which may be of different materials. Determination for the correct physics locally necessarily requires the contact problem.

4.1.2 The Mathematical Problem

Entire textbooks are devoted to contact mechanics; therefore, only a brief overview is presented here. [**?**]

4.2 Elastodynamics

The cold spray process is necessarily a dynamic one, so one must resolve the physics incrementally.

4.2.1 Linear Elasticity

One may model the cold spray process as an elastic-plastic problem, though the elasticity portion is far less significant.

The strong form of linear elasticity is given by Cauchy's equations.

$$\rho \ddot{x} = \nabla \cdot \sigma + f \tag{15}$$

$$\sigma = \lambda \, \mathrm{tr}(\varepsilon)\mathbf{I} + 2\mu\epsilon \tag{16}$$

$$\varepsilon = \frac{1}{2}(\nabla x + (\nabla x)^T) \tag{17}$$

We will now convert this into the weak form that will be more useful for us which will be elaborated upon in the next subsection. v will be our test functions.

$$\int \rho \ddot{x} \cdot v = \int (\nabla \cdot \sigma) \cdot v + \int f \cdot v \tag{18}$$

$$\int \rho \ddot{x} \cdot v + \int \sigma(x) : \varepsilon(v) = \int f \cdot v \tag{19}$$

By application of the divergence theorem on the internal stress term.

We expand the stress and strain tensor with the constitutive equations **??** and **??**.

$$\int \rho \ddot{x} \cdot v + \int \lambda(\boldsymbol{\nabla} \cdot x)(\boldsymbol{\nabla} \cdot v) + 2\mu\epsilon(x) : \epsilon(v) = \int f \cdot v \tag{20}$$

Naturally, we convert the equation into Galerkin's formulation to then input into FreeFEM. Einstein summation convention here will be used.

$$x(t) = X_i(t)u_i(x) \tag{21}$$

$$\rho \ddot{X}_i \int u_i \cdot v_j + X_i \int \lambda(\boldsymbol{\nabla} \cdot u_i)(\boldsymbol{\nabla} \cdot v_j) + 2\mu\varepsilon(u_i) : \varepsilon(v_j) = \int f \cdot v_j \tag{22}$$

Observe that this gives us the following equation.

$$M_{ij}\ddot{x}_j + K_{ij}x_j = f_i \tag{23}$$

$$M_{ij} = \int u_j \cdot v_i \; ; \; K_{ij} = \int \lambda(\boldsymbol{\nabla} \cdot u_j)(\boldsymbol{\nabla} \cdot v_i) + 2\mu\varepsilon(u_j) : \varepsilon(v_i) \; ; \; f_i = \int f \cdot v_i \tag{24}$$

Which is the form to then plug into the α integrators such as HHT-α and Newark-β.

4.2.2 Newark Beta

Here we will present the incremental Newark Beta integrator for structural dynamics.[?]

We follow Gavin, H.'s lecture notes on incremental formulation of the Newark integrator.

The Newark integrator begins with the following assumptions on the form of the differential equation and the discretizations.

$$M\ddot{x} + C\dot{x} + Kx + R(x, \dot{x}) = f \tag{25}$$

$$\dot{x}_{i+1} \approx \dot{x}_i + h\left[(1-\gamma)\ddot{x}_i + \gamma\ddot{x}_{i+1}\right] \tag{26}$$

$$x_{i+1} \approx x_i + h\dot{x}_i + h^2\left[\left(\frac{1}{2} - \beta\right)\ddot{x}_i + \beta\ddot{x}_{i+1}\right] \tag{27}$$

Where M is the mass matrix, C the dissipation matrix, K the stiffness matrix, R will be any nonlinear components, f being any external forces, and A_i being $A(t = ih)$.

The free parameters β and γ are left to be tuned for desired integrator behaviors. Choices such as $\gamma < 1/2$, the integrator is conditionally stable, whilst the chosen method for this project is implicit and unconditionally stable with $\beta = 1/4$ and $\gamma = 1/2$.

We now solve for $\ddot{x}_{i+1}$.

$$\ddot{x}_{i+1} = \frac{1}{\beta h^2}(x_{i+1} - x_i) - \frac{1}{\beta h}\dot{x}_i - \left(\frac{1}{2\beta} - 1\right)\ddot{x}_i \tag{28}$$

Giving us this formula for the next velocity.

$$\dot{x}_{i+1} = \frac{\gamma}{\beta h}(x_{i+1} - x_i) - \left(\frac{\gamma}{\beta} - 1\right)\dot{x}_i + h\left(1 - \frac{\gamma}{2\beta}\right)\ddot{x}_i \tag{29}$$

4.2.3 Numerical FreeFEM Implementation

We consider a linear elastic block resting on a table, slumping under the effect of gravity. The constant parameters are as follows.

Name	Notation in Code	Value
Gravity	g	-9.8 m/s^2
Density	ρ	1
Young's modulus	E	1E3 $kg/m/s^2$
Poisson modulus	nu	0.3
Newark parameter	beta	0.25
Newark parameter	gamma	0.5
Rayleigh dissipation parameter	AlphaM	0.01
Rayleigh dissipation parameter	AlphaK	0.01

We introduce some notation to simplify the equations **??** and **??**.

$$c_0 = \frac{1}{\beta h^2} \ ; \ c_1 = \frac{1}{\beta h} \ ; \ c_2 = \frac{1}{2\beta} - 1 \tag{30}$$

$$c_3 = \frac{\gamma}{\beta h} \ ; \ c_4 = \frac{\gamma}{\beta} - 1 \ ; \ c_5 = h\left(1 - \frac{\gamma}{2\beta}\right) \tag{31}$$

As shown in the code below.

```
// Beta Newmark constants
real c0 = 1./(beta*dt*dt);
real c1 = 1./(beta*dt);
real c2 = 1./(2.*beta) - 1.;
real c3 = gamma/(beta*dt);
real c4 = gamma/beta - 1.;
real c5 = dt*(gamma/(2*beta) - 1.);
```

The Lame parameters for the stress tensor are given as follows.

$$\lambda = \frac{E\nu}{(1+\nu)(1-2\nu)} \ ; \ \mu = (3 * 10^{-2})\frac{E}{2(1+\nu)} \tag{32}$$

Where we have chosen to scale the 2nd Lame parameter by 3e-2 to amplify visual effects.

A particular sanity check is that under dampening, the system will converge to the steady state, which one can trivially compute. So, we add in the Rayleigh dissipation terms into the dissipation matrix C.

$$C = \alpha_M M + \alpha_K K \tag{33}$$

Here we introduce some more notations to simplify the Galerkin formulation.

$$M(a) = \int a \cdot v \ ; \ K(a) = \int \lambda(\boldsymbol{\nabla} \cdot a)(\boldsymbol{\nabla} \cdot v) + 2\mu\varepsilon(a) : \varepsilon(v) \tag{34}$$

As shown in the code below.

```
// Macros M and K as described in the PDF
macro M(a) (a[0]*pvi+a[1]*pvj) //
macro K(a) (lambda*(div(a)*div([pvi,pvj]))+2*muK*mu*eps(a)'*
    eps([pvi,pvj])) //
```

Altogether, we form the Galerkin formulation.

$$\begin{aligned}
M\ddot{x}_{i+1} + C\dot{x}_{i+1} + Kx_{i+1} - f_{i+1} = 0 \\
M(c_0(x_{i+1} - x_i) - c_1\dot{x}_i - c_2\ddot{x}_i) + \\
(\alpha_M M + \alpha_K K)(c_3(x_{i+1} - x_i) - c_4\dot{x}_i - c_5\ddot{x}_i) + \\
Kx_{i+1} - f_{i+1} = 0 \\
c_0 M(x_{i+1}) - c_0 M(x_i) - c_1 M(\dot{x}_i) - c_2 M(\ddot{x}_i) + \\
\alpha_M(c_3 M(x_{i+1}) - c_3 M(x_i) - c_4 M(\dot{x}_i) - c_5 M(\ddot{x}_i)) + \\
\alpha_K(c_3 K(x_{i+1}) - c_3 K(x_i) - c_4 K(\dot{x}_i) - c_5 K(\ddot{x}_i)) + \\
K(x_{i+1}) - f_{i+1=0}
\end{aligned}$$

Finally, we shuffle terms to have all additions in 1 integral and all subtractions in another.

$$
\begin{aligned}
c_0 M(x_{i+1}) + \alpha_M c_3 M(x_{i+1}) + \alpha_K c_3 K(x_{i+1}) + K(x_{i+1}) \\
- \\
c_0 M(x_i) + c_1 M(\dot{x}_i) + c_2 M(\ddot{x}_i) + \\
\alpha_M (c_3 M(x_i) + c_4 M(\dot{x}_i) + c_5 M(\ddot{x}_i)) + \\
\alpha_K (c_3 K(x_i) + c_4 K(\dot{x}_i) + c_5 K(\ddot{x}_i)) + \\
f_{i+1} = 0
\end{aligned}
$$

As shown in the code.

```
problem BetaNewark([px,py],[pvi,pvj])
= int2d(Ph)(
        c0*M([px,py])+
                AlphaM*c3*M([px,py])+
                AlphaK*c3*K([px,py])+
                K([px,py])
        )
- int2d(Ph)(
                c0*M([un,vn])+
                c1*M([uvn,vvn])+
                c2*M([uan,van])+
                AlphaM*(c3*M([un,vn])+c4*M([uvn,vvn])+c5*M([
                    uan,van]))+
                AlphaK*(c3*K([un,vn])+c4*K([uvn,vvn])+c5*K([
                    uan,van]))+
                g*pvj
        )
+ on(bottom, py=0);
```

Using FreeFEM's builtin hTriangle function, we can obtain the size of the finite elements. This way we can adequately choose the time step size for a good CFL number. The minimum size element in this case has the size of 0.0820061m.

Wave Type	Speed	Time Step for CFL=1
Pressure	24.4949 m/s	0.00334788s
Shear	3.39683 m/s	0.0241419s

We choose the Eulerian description and do not adapt the meshes here in the toy problem, therefore the mesh sizes never changes, so we choose a time step of 0.00334788s. Due to the choice of Newark parameters, no stability issues will be present. The specific choice of time step is for better convergence.

4.2.4 Results

The results can be viewed as a time series in Paraview, or any .vtu viewer. The block jiggles and the movements slowly decay away.

4.3 FreeFEM Implementation

4.3.1 Physical Model

4.3.2 Results

5 Plasticity

5.1 The Elastoplasticity Problem

5.1.1 Description and Connection with Cold Spray

Plasticity is the study of materials that undergo irreversible, permanent deformations under large loads.

Cold spray deposition results from micron sized metal particles impacting a target substrate at supersonic speeds. Plastic and thermomechanical effects are the dominant physics in cold spray deposition. Under the viscoplastic deformation and thermomechanical behaviors, the metal particles undergo shear thinnign then metallurgically binds the target substrate. Understanding the plastic processes is necessary for modeling cold spray deposition.

5.1.2 The Mathematical Problem

Plasticity is inherently an experimental science. The mathematical theories of plasticity are formulated to represent experimental findings and physical models, and such is more phenomenological. The physical models of plasticity is concerned with how plasticity occurs.

The physical material of interest will completely dictate inelastic processes. The materials of interest can vary greatly from fairly regular metals, to soil types, and even rocks and concrete. Respectively, some modes of fault: slip bands, compactification of soil or voids and shearing, and opening and closing of cracks. [**?**]

To describe the onset and accumulations of permanent deformation, known as yielding, we need a criterion for the transition from elastic to plastic deformation. From physical experiments, the transition from elastic to plastic may not be obvious, if such a point even exists in reality. There are then 2 main types of constitive models for plastic deformation. One assuming the existence of a yield point, and one that does not. Here we present the former. [**?**]

Suppose there exists a function of all internal state variables called the yield function – $F(q)$. We denote the collection of all state variables with q. We assume the following.

$$\text{if } F(q) < 0, \text{ elastic deformation for all such q. Plastic otherwise.} \tag{35}$$

Naturally, the manifold given by $F(q) = 0$ represent the point of transition from elastic to plastic deformation. The manifold is known as the yield surface.

The specific $F(q)$ will depend on the model.

5.2 FreeFEM Implementation

5.2.1 Physical Model

We concern ourselves with the model of a 50um by 75um block of copper impacting a rigid wall at 500m/s.

We proceed with the Johnson-Cook model of plasticity. [?]

The yield function of the Johnson-Cook plasticity model is given by the difference of the von Mises stress and the Johnson-Cook yield stress.

$$f(q) = \sigma_{vm} - \sigma_Y \tag{36}$$

$$\sigma_{vm} = \sqrt{\frac{3}{2} s : s} \ ; \ s = \sigma - \frac{1}{3}\operatorname{Tr}(\sigma) I \tag{37}$$

$$\sigma_Y = (A + B\bar{\epsilon}_p^{n_{JC}})(1 + C\ln(\dot{\bar{\epsilon}}_p/\dot{\epsilon}_0))(1-\theta^m) \ ; \ \theta = \frac{T - T_0}{T_m - T_0} \tag{38}$$

Where σ_{vm} is the von Mises stress.

Now with the

For the accumulation of plastic deformation here, it will be given by the flow rule and Drucker postulate of associative flow.

$$\dot{\epsilon}_p = \dot{\lambda}\frac{\partial F(q)}{\partial \sigma} = \dot{\lambda}\frac{3J_2}{\sigma_{vm}} \tag{39}$$

$\dot{\lambda}$ is a proportionality parameter that needs to be determined. Fortunately for us, with von Mises' $J2$ based associated flow plasticity, $\dot{\lambda}$ is equivalent to $\dot{\bar{\epsilon}}_p$.

Johnson-Cook is also a thermomechanical model, so the energy delivered via the plastic process is given by the following equation.

$$\dot{W}_p = \sigma : \dot{\epsilon}_p \tag{40}$$

By now in our numerical algorithm, we have forced $F(q) = 0 = \sigma_{vm} - \sigma_Y$.

5.2.2 Elastic Portion only

We set up the model as a linear elastic problem, integrated over time to 50ns using Newark beta.

To avoid the cost of a contact algorithm, we idealize the circumstances and initiate the model right when the block hits the wall, and enforce a Dirichlet condition such that it doesn't penetrate it.

5.2.3 Plasticity with Linear Elasticity

We now need to determine the equivalent plastic strain increment from equation **??**. The determination of $\dot{\lambda}$ can be done numerically, and it is likely going to be some form of nonlinear optimization problem. We can apply Karush-Kuhn-Tucker (KKT) if we enforce a few criterions that are also sensible physics. KKT does the following with some restrictions.[?]

$$\text{minimize } f(x) \tag{41}$$
$$\text{subject to } g_i(x) \leq 0 \text{ ; } h_j(x) = 0 \tag{42}$$

We can force $F(q) \leq 0$ and still maintain the flow rule and obtain the necessary restrictions for KKT to apply. Recall the assumptions in the yield criterion **??**. If the process is elastic – $F(q) < 0$, the plastic strain increment is 0, so in those cases $\dot{\lambda} = 0$. If the process is elastoplastic – $F(q) = 0$ under the new restriction, the plastic strain increment is > 0, so $\dot{\lambda} > 0$. By enforcing a new condition $F(q) \leq 0$, we can obtain the needed extra h_j condition which we may notice is $F(q)\dot{\lambda} = 0$.

Giving us the following conditions for applying KKT.

$$\dot{\lambda} \geq 0 \tag{43}$$
$$F(q) \leq 0 \tag{44}$$
$$F(q)\dot{\lambda} = 0 \tag{45}$$

5.2.4 Results

We first present the elastic results. At the end of those 50ns, we already see the block begin to rebound off the wall. We are definitely accumulating errors due to not using a contact algorithm.

6 Cold Spray

6.1 Overview

6.2 FreeFEM Implementation

6.2.1 Results

results/StefanIce0.png

(a) Ice and water at the start

results/StefanIce1.png

(b) Ice and water at the end

results/Stefan0.png

(a) t=0s

results/Stefan1.png

(b) t=7.5s

results/Stefan2.png

results/Beta0.png

(a) t=0s

results/Beta1.png

(b) t=0.75s

results/Beta2.png